Pendulum in cartesian coordinates

RK based numerical scheme

```
> restart:
```

Pendulum equation reduced to first order mass=1

```
> EQ1 := diff(x(t),t)-u(t) ;
  EQ2 := diff(y(t),t)-v(t) ;
  EQ3 := diff(u(t),t)+x(t)*mu(t) ;
  EQ4 := diff(v(t),t)+y(t)*mu(t)+g ;
  ALG := x(t)^2+y(t)^2-1 ;
```

$$EQ1 := \frac{d}{dt} x(t) - u(t)$$

$$EQ2 := \frac{d}{dt} y(t) - v(t)$$

$$EQ3 := \frac{d}{dt} u(t) + x(t)\,\mu(t)$$

$$EQ4 := \frac{d}{dt} v(t) + y(t)\,\mu(t) + g$$

$$ALG := x(t)^2 + y(t)^2 - 1 \qquad \textbf{(1)}$$

Derivate constraint two times

```
> DALG   := diff(ALG,t);
  DDALG  := diff(DALG,t);
```

$$DALG := 2\,x(t)\left(\frac{d}{dt} x(t)\right) + 2\,y(t)\left(\frac{d}{dt} y(t)\right)$$

$$DDALG := 2\left(\frac{d}{dt} x(t)\right)^2 + 2\,x(t)\left(\frac{d^2}{dt^2} x(t)\right) + 2\left(\frac{d}{dt} y(t)\right)^2 + 2\,y(t)\left(\frac{d^2}{dt^2} y(t)\right) \qquad \textbf{(2)}$$

Stabilize constraint with Baumgarte (reduce DAE index 3 to DAE index 1)

```
> ALGSTAB := DDALG + 2*zeta*omega*DALG + omega^2 * ALG ;
```

$$ALGSTAB := 2\left(\frac{d}{dt} x(t)\right)^2 + 2\,x(t)\left(\frac{d^2}{dt^2} x(t)\right) + 2\left(\frac{d}{dt} y(t)\right)^2 + 2\,y(t)\left(\frac{d^2}{dt^2} y(t)\right) + 2\,\zeta\,\omega\left(2\,x(t)\left(\frac{d}{dt} x(t)\right) + 2\,y(t)\left(\frac{d}{dt} y(t)\right)\right) + \omega^2\left(x(t)^2 + y(t)^2 - 1\right) \qquad \textbf{(3)}$$

Solve for velocity and acceleration

```
> RESVEL := solve( {EQ1,EQ2}, diff({x(t),y(t)},t) ) ;
  RESACC := solve( {EQ3,EQ4}, diff({u(t),v(t)},t) ) ;
```

$$RESVEL := \left\{\frac{d}{dt} x(t) = u(t),\ \frac{d}{dt} y(t) = v(t)\right\}$$

$$RESACC := \left\{\frac{d}{dt} u(t) = -x(t)\,\mu(t),\ \frac{d}{dt} v(t) = -y(t)\,\mu(t) - g\right\} \qquad \textbf{(4)}$$

```
> ALGSTAB := subs(RESACC, subs( RESVEL,ALGSTAB)) ;
```

(5)

$$ALGSTAB := 2\,u(t)^2 - 2\,x(t)^2\,\mu(t) + 2\,v(t)^2 + 2\,y(t)\,(-y(t)\,\mu(t) - g) + 2\,\zeta\,\omega\,(2\,x(t)\,u(t) + 2\,y(t)\,v(t)) + \omega^2\,(x(t)^2 + y(t)^2 - 1) \quad \textbf{(5)}$$

Solve the stabilized equation for mu

```
> RESMU := collect(solve( ALGSTAB, {mu(t)} ),omega) ;
```

$$RESMU := \left\{\mu(t) = \frac{1}{2}\,\frac{(x(t)^2 + y(t)^2 - 1)\,\omega^2}{x(t)^2 + y(t)^2} + \frac{1}{2}\,\frac{(4\,\zeta\,x(t)\,u(t) + 4\,\zeta\,y(t)\,v(t))\,\omega}{x(t)^2 + y(t)^2} + \frac{1}{2}\,\frac{2\,u(t)^2 + 2\,v(t)^2 - 2\,y(t)\,g}{x(t)^2 + y(t)^2}\right\} \quad \textbf{(6)}$$

Pendulum DAE reduced to index 1 and stabilized with Baumgarte

```
> EQ||(1..4) ;
  EQ5 := mu(t)-subs(RESMU,mu(t)) ;
```

$$\frac{d}{dt}\,x(t) - u(t),\ \frac{d}{dt}\,y(t) - v(t),\ \frac{d}{dt}\,u(t) + x(t)\,\mu(t),\ \frac{d}{dt}\,v(t) + y(t)\,\mu(t) + g$$

$$EQ5 := \mu(t) - \frac{1}{2}\,\frac{(x(t)^2 + y(t)^2 - 1)\,\omega^2}{x(t)^2 + y(t)^2} - \frac{1}{2}\,\frac{(4\,\zeta\,x(t)\,u(t) + 4\,\zeta\,y(t)\,v(t))\,\omega}{x(t)^2 + y(t)^2} - \frac{1}{2}\,\frac{2\,u(t)^2 + 2\,v(t)^2 - 2\,y(t)\,g}{x(t)^2 + y(t)^2} \quad \textbf{(7)}$$

Now we can use any numericak methods for ODE, for example Collatz
x'=f(x,t)

x(k+1/2) = x(k) + (DT/2) * f(x(k),t(k));
x(k+1) = x(k) + DT*f(x(k+1/2),t(k+1/2)) ;

First HALF step

```
> SUBSH := [ diff(x(t),t) = (xH-xO)/DT,
             diff(y(t),t) = (yH-yO)/DT,
             diff(u(t),t) = (uH-uO)/DT,
             diff(v(t),t) = (vH-vO)/DT,
             x(t) = xO,
             y(t) = yO,
             u(t) = uO,
             v(t) = vO,
             mu(t) = muO ];
```

$$SUBSH := \left[\frac{d}{dt}\,x(t) = \frac{xH - xO}{DT},\ \frac{d}{dt}\,y(t) = \frac{yH - yO}{DT},\ \frac{d}{dt}\,u(t) = \frac{uH - uO}{DT},\ \frac{d}{dt}\,v(t) = \frac{vH - vO}{DT},\ x(t) = xO,\ y(t) = yO,\ u(t) = uO,\ v(t) = vO,\ \mu(t) = muO\right] \quad \textbf{(8)}$$

```
> MUH := solve( subs(SUBSH, EQ5), {muO} ) ;
```

$$MUH := \left\{muO = \frac{1}{2}\,\frac{\omega^2\,xO^2 + \omega^2\,yO^2 - \omega^2 + 4\,\zeta\,\omega\,xO\,uO + 4\,\zeta\,\omega\,yO\,vO + 2\,uO^2 + 2\,vO^2 - 2\,yO\,g}{xO^2 + yO^2}\right\} \quad \textbf{(9)}$$

```
> subs(SUBSH,[EQ||(1..4)]);
```

$$\left[\frac{xH-xO}{DT}-uO,\ \frac{yH-yO}{DT}-vO,\ \frac{uH-uO}{DT}+xO\,muO,\ \frac{vH-vO}{DT}+yO\,muO+g\right] \qquad (10)$$

```
> HSTEP := subs(MUH,op(solve( subs(SUBSH,[EQ||(1..4)]), [xH,yH,uH,
vH] ))) ;
```

$$HSTEP := \Big[xH=xO+uO\,DT,\ yH=yO+vO\,DT,\ uH=uO - \frac{1}{2}\,\frac{1}{xO^2+yO^2}\Big(xO\big(\omega^2\,xO^2+\omega^2\,yO^2-\omega^2+4\,\zeta\,\omega\,xO\,uO+4\,\zeta\,\omega\,yO\,vO + 2\,uO^2+2\,vO^2-2\,yO\,g\big)\,DT\Big),\ vH=vO - \frac{1}{2}\,\frac{1}{xO^2+yO^2}\Big(yO\big(\omega^2\,xO^2+\omega^2\,yO^2-\omega^2+4\,\zeta\,\omega\,xO\,uO+4\,\zeta\,\omega\,yO\,vO + 2\,uO^2+2\,vO^2-2\,yO\,g\big)\,DT\Big)-g\,DT\Big] \qquad (11)$$

Second FULL step

```
> SUBSF := [ diff(x(t),t) = (xN-xO)/DT,
             diff(y(t),t) = (yN-yO)/DT,
             diff(u(t),t) = (uN-uO)/DT,
             diff(v(t),t) = (vN-vO)/DT,
             x(t) = xH,
             y(t) = yH,
             u(t) = uH,
             v(t) = vH,
             mu(t) = muH ];
```

$$SUBSF := \left[\frac{d}{dt}\,x(t)=\frac{xN-xO}{DT},\ \frac{d}{dt}\,y(t)=\frac{yN-yO}{DT},\ \frac{d}{dt}\,u(t)=\frac{uN-uO}{DT},\ \frac{d}{dt}\,v(t)=\frac{vN-vO}{DT},\ x(t)=xH,\ y(t)=yH,\ u(t)=uH,\ v(t)=vH,\ \mu(t)=muH\right] \qquad (12)$$

```
> MUN := solve( subs(SUBSF, EQ5), {muH} ) ;
```

$$MUN := \left\{muH=\frac{1}{2}\,\frac{\omega^2\,xH^2+\omega^2\,yH^2-\omega^2+4\,\zeta\,\omega\,xH\,uH+4\,\zeta\,\omega\,yH\,vH+2\,uH^2+2\,vH^2-2\,yH\,g}{xH^2+yH^2}\right\} \qquad (13)$$

```
> FSTEP := subs( MUN, op(solve( subs(SUBSF,[EQ||(1..4)]), [xN,yN,
uN,vN] ))) ;
```

$$FSTEP := \Big[xN=xO+uH\,DT,\ yN=yO+vH\,DT,\ uN=uO - \frac{1}{2}\,\frac{1}{xH^2+yH^2}\Big(xH\big(\omega^2\,xH^2+\omega^2\,yH^2-\omega^2+4\,\zeta\,\omega\,xH\,uH+4\,\zeta\,\omega\,yH\,vH \qquad (14)$$

$$+ 2\, uH^2 + 2\, vH^2 - 2\, yH\, g\big)\, DT\Big),\; vN = vO$$

$$- \frac{1}{2}\, \frac{1}{xH^2 + yH^2} \Big(yH \big(\omega^2\, xH^2 + \omega^2\, yH^2 - \omega^2 + 4\, \zeta\, \omega\, xH\, uH + 4\, \zeta\, \omega\, yH\, vH$$

$$+ 2\, uH^2 + 2\, vH^2 - 2\, yH\, g\big)\, DT\Big) - g\, DT\Big]$$

```
> advance := proc ( x0, y0, u0, v0, dt, N )
    local kk, SUBS, SUBSBASE, x1, y1, u1, v1, xh, yh, uh, vh, XY,
  UV ;
    XY := [[x0,y0]] ;
    UV := [[u0,v0]] ;
    SUBSBASE := { g=9.81, DT=dt,zeta=1, omega=1 } ;
    for kk from 1 to N do
      # Half Step
      SUBS := SUBSBASE union { xO=XY[-1][1], yO=XY[-1][2],
                               uO=UV[-1][1], vO=UV[-1][2]} ;
      xh, yh, uh, vh := op(evalf(subs( SUBS, subs( HSTEP, [xH,yH,
  uH,vH] )))) ;


      # Full Step
      SUBS := SUBS union { g=9.81, DT=dt, xH=xh, yH=yh,uH=uh, vH=
  vh } ;
      x1, y1, u1, v1 := op(evalf(subs( SUBS, subs( FSTEP, [xN,yN,
  uN,vN] )))) ;

      XY  := [op(XY),[x1,y1]] ;
      UV  := [op(UV),[u1,v1]] ;
    end  ;
    [XY,UV] ;
  end proc:
```

Test numerical scheme DT = 1/200

```
> RES := advance( 1, 0, 0, 0, 10/2000, 4000 ) :
> plot( RES[1] ) ;
```



Test numerical scheme DT = 1/20

```
> RES := advance( 1, 0, 0, 0, 10/200, 2000 ) :
  plot( RES[1] ) ;
```


−0,6
−0,4
−0,2
0
0,2
0,4
0,6
0,8
1,0
−0,2
−0,4
−0,6
−0,8
−1,0

>

>